

The BENCHMARKING Procedure

Overview

This procedure will ensure coherence between time series data of the same target variable measured at different frequencies, e.g. sub-annually and annually. Benchmarking consists of imposing the level of the benchmark series while minimizing the revisions of the observed movement in the sub-annual series as much as possible. The procedure also allows nonbinding benchmarking where the benchmark series can also be revised.

The procedure may also be used for benchmarking-related topics such as temporal distribution (disaggregation of the benchmark series into more frequent observations), calendarization (special case of temporal distribution) and linkage (connection of different time series segments into a consistent single time series).

Procedure Syntax

PROC BENCHMARKING <option(s)>;

VAR var1 </alt1> <...varN </altN> >;

WITH var1 </alt1> <...varN </altN> >;

BY variable(s);

To do this	Use this statement
Perform benchmarking separately for each BY group	BY
Identify, as needed, the sub-annual series to benchmark along with their associated alterability coefficients	VAR
Identify, as needed, the benchmark series along with their associated alterability coefficients	WITH

PROC BENCHMARKING Statement

PROC BENCHMARKING <option(s)>;

To do this	Use this option
Specify the input benchmarks data set	BENCHMARKS=
Specify a value for the bias	BIAS=
Specify the bias estimation option	BIASOPTION=
Specify the value for the λ (lambda) parameter	LAMBDA=
Specify the output benchmarks data set	OUTBENCHMARKS=
Specify the output data set that will contain supplementary data useful to produce analytical tables and graphs	OUTGRAPHTABLE=
Specify the output benchmarked sub-annual series data set	OUTSERIES=

Specify the value for the ρ (rho) parameter	RHO=
Specify the input sub-annual series data set	SERIES=
Specify the tolerance level for the ultimate test in percentage or absolute value	TOLERANCEPERCENT= TOLERANCEVALUE=
Specify the threshold for identification of negative values.	TOLNEGRESULT=
Print information about CPU time and print intermediate results	VERBOSE
Display a warning message in the log when a negative value is found (see option TOLNEGRESULT=)	WARNNEGRESULT NOWARNNEGRESULT

Options

BENCHMARKS=*SAS-data-set*

specifies the input SAS data set that contains the benchmarks. It is mandatory. The following numeric variables define the coverage period of each benchmark and must be in this data set: STARTYEAR, STARTPERIOD, ENDYEAR and ENDPERIOD. In addition, a numeric variable named VALUE containing the values of the benchmark must be present by default. To specify another variable name or more than one benchmark series, use the VAR and/or WITH statements.

BIAS=*real number*

specifies the value of the user-defined bias to be used for the correction of the sub-annual series prior to benchmarking. The bias is added to the sub-annual series with an additive model (when LAMBDA=0.0) while it is multiplied otherwise (when LAMBDA \neq 0.0). Option BIAS= is not mandatory. The default value is BIAS=0.0 when LAMBDA=0.0 and BIAS=1.0 otherwise. Note that option BIAS= is not used when option BIASOPTION=3 (see below).

BIASOPTION=*1, 2 or 3*

specifies the bias estimation option. It is mandatory.

If you specify this bias estimation option value	Then the procedure
1	Will not estimate the bias parameter. The bias used to correct the sub-annual series will be the value specified with the BIAS= option or the default value (see option BIAS=).
2	Will estimate the bias parameter, print the result to the log, but will not use it. The bias used to correct the sub-annual series will be the value specified with the BIAS= option or the default value (see option BIAS=).
3	Will estimate the bias parameter, print the result to the log and apply the estimated bias correction. Any value specified with the BIAS= option will be printed to the log but not used.

LAMBDA=*real number (suggested values are between -3 and 3)*

specifies the value of the adjustment model parameter λ . This option is mandatory. Typical values are LAMBDA=0.0 for an additive model and LAMBDA=1.0 for a proportional model. Prorating can be obtained by setting LAMBDA=0.5 and RHO=0.0.

OUTBENCHMARKS=SAS-data-set

names the output SAS data set that will contain the benchmarks used by the procedure. If it is not specified, PROC BENCHMARKING will create it by using the DATA n naming convention. The variables of this data set will have the same names as the ones in the BENCHMARKS= input data set. If BY variables are specified, the BY variables will also appear on this data set.

OUTGRAPHTABLE=SAS-data-set

names the output SAS data set that will contain supplementary data useful to produce analytical tables and graphs in the case of non-overlapping benchmarks. It is optional and is not created if not specified. The following table describes the variables that will be written to the data set. If BY variables are specified, they will also appear on this data set.

Variable Name	Description
ALTBENCHMARKS	Name of the variable used to specify the benchmark series alterability coefficients
ALTBENCHMARKSVALUE	The benchmark series alterability coefficient values
ALTSERIES	Name of the variable used to specify the sub-annual series alterability coefficients
ALTSERIESVALUE	The sub-annual series alterability coefficient values
AVGBENCHMARK	Value of the benchmark divided by the number of covered periods; missing if there is no benchmark
AVGBENCHMARKSUBANNUALRATIO	Ratios ^{**} of the values of variables AVGBENCHMARK and AVGSUBANNUAL; missing if there is no benchmark
AVGSUBANNUAL	Values of the initial sub-annual series averaged over the benchmark coverage period; missing if there is no benchmark
BENCHMARKED	Benchmarked values of the sub-annual series
BENCHMARKEDSUBANNUALRATIO	Ratios ^{**} of the values of variables BENCHMARKED and SUBANNUAL
BIAS	The value of BIAS used (calculated by the procedure or supplied by user)
DATE	Character variable combining the values of variables YEAR and PERIOD
GROWTHRATEBENCHMARKED	Growth rates ^{**} in benchmarked sub-annual series.
GROWTHRATESUBANNUAL	Growth rates ^{**} in initial sub-annual series
LAMBDA	The value of the input parameter λ (lambda)
M	The benchmark coverage period identifiers, $m = 1, \dots, M$; missing if there is no benchmark
PERIOD	Period values
PERIODICITY	The maximum number of periods in a year, for example 4 for a quarterly series
RHO	The value of the input parameter ρ (rho)
SUBANNUAL	Sub-annual series initial values
SUBANNUALCORRECTED	Sub-annual series rescaled values ('corrected for bias')
T	The sub-annual period identifiers, $t = 1, \dots, T$

VARBENCHMARKS	Name of the variable used to specify the benchmark series
VARSERIES	Name of the variable used to specify the sub-annual series
YEAR	Year values

** Terms “ratios” and “growth rates” refer here to a terminology commonly used in time series and actually mean “differences” and “first differences” respectively when the additive model is used (LAMBDA=0.0).

OUTSERIES=*SAS-data-set*

names the output SAS data set that will contain the benchmarked sub-annual series. If it is not specified, PROC BENCHMARKING will create it by using the DATA*n* naming convention. The variables of this data set will have the same names as the ones in the SERIES= input data set. If BY variables are specified, they will also appear on this data set.

RHO=*real number between 0 and 1 (inclusive)*

specifies the value of the autoregressive parameter ρ in the $[0,1]$ interval. It is mandatory.

SERIES=*SAS-data-set*

specifies the input SAS data set that contains the sub-annual series to benchmark. It is mandatory. The following numeric variables must be in this data set: YEAR and PERIOD. In addition, a numeric variable named VALUE containing the values of the sub-annual series to be processed must be present by default. To specify another variable name or more than one sub-annual series, use the VAR statement.

TOLERANCEPERCENT=
TOLP=*positive real number (including 0)* | **TOLERANCEVALUE**=
TOLV=*positive real number (including 0)*

specifies the tolerance, in percentage or absolute value, to be used when performing the ultimate test in the case of binding benchmarks (alterability coefficient = 0.0 for the benchmarks). The test compares the input annual benchmarks with benchmarks calculated from the benchmarked series. This number is optional.

Default:	TOLERANCEVALUE=1E-3
Example:	To set a tolerance of 1%, specify TOLERANCEPERCENT=0.01 To set a tolerance of 10, specify TOLERANCEVALUE=10

TOLNEGRESULT=
TOLN=*strictly negative real number (excluding 0)*

specifies the threshold for identification of negative values. A value is considered negative when it is smaller than this threshold. This number is optional. The default value for this option is -1E-3.

VERBOSE

use this option to tell the procedure to print intermediate results and CPU time. With this option, intermediate matrices will be printed to the log as well as the CPU time taken to process them. (Inverse of matrices, multiplication of matrices, etc.).

WARNNEGRESULT | NOWARNNEGRESULT

specifies whether PROC BENCHMARKING displays a warning message in the log when a negative value created by the procedure is smaller than the threshold specified by the TOLNEGRESULT= option. The default option is WARNNEGRESULT.

VAR Statement

VAR var1 </alt1> <...varN </altN> >;

Required arguments

var1...varN

specifies the variable(s) containing the values of the sub-annual series to be benchmarked. The variables must be numeric. The VAR statement is optional. If not specified, the procedure will look for a variable called VALUE in the SERIES= input data set.

alt1...altN

specifies, as needed, the variable(s) containing the alterability coefficients that apply to the sub-annual series. The alterability coefficients variables should be numeric. If an alterability coefficients variable is not specified for a given sub-annual series, the procedure will use the default alterability coefficient value for sub-annual series, which is 1.0. Alterability coefficients only come into play after the initial sub-annual series has been corrected for the bias (when applicable). For example, this means that specifying an alterability coefficient of 0.0 for a given sub-annual series data point *will not* result in an unchanged value after benchmarking if bias correction is used (see options BIAS= and BIASOPTION=). Specification of alterability coefficients is not allowed when RHO=1.

WITH Statement

WITH var1 </alt1> <...varN </altN> >;

Required arguments

var1...varN

specifies the variable(s) that contain the values of the benchmarks. The variables must be numeric. The WITH statement is optional. If not specified, the procedure will look in the BENCHMARKS= input dataset for either the variable VALUE or variables corresponding to the names specified in the VAR statement.

alt1...altN

specifies, as needed, the variable(s) containing the alterability coefficients that apply to the benchmarks. The alterability coefficients variables should be numeric. If an alterability coefficients variable is not specified for a given benchmark, the procedure will use the default alterability coefficient value for a benchmark, which is 0.0. Specification of alterability coefficients is not allowed when RHO=1.

BY Statement

BY *variable-1* <... *variable-n*>;

Required arguments

variable(s)

specifies the variables used by the procedure to form BY groups. The BY variables must be present in the two input data sets (BENCHMARKS= and SERIES=) and will appear in all three output data sets (OUTBENCHMARKS=, OUTGRAPHTABLE and OUTSERIES=). You can specify more than one variable. This statement is optional. The variables can be numeric or character.

Details

- If a missing value appears in one of the variables of the BENCHMARKS= input data set (other than the BY variables), the observations with the missing values are dropped, a warning message is displayed in the log and the procedure runs.
- If a missing value appears in the YEAR and/or PERIOD variables of the SERIES= input data set and BY variables are specified, the corresponding BY group is skipped, a warning message is displayed in the log and the procedure moves on to the next BY group. If no BY variables are specified, a warning message is displayed in the log and no processing is done at all.
- If a missing value appears in a variable identifying a sub-annual series variables in the SERIES= input data set (see the VAR statement) and BY variables are specified, the corresponding BY group is skipped, a warning message is displayed in the log and the procedure moves on to the next BY group. If no BY variables are specified, the affected sub-annual series is not processed, a warning message is displayed in the log and the procedure moves on to the next sub-annual series (when applicable).
- The procedure does not allow proportional benchmarking (LAMBDA≠0) of a sub-annual series that is null over the entire coverage period of a non-null benchmark.
- With a value of RHO=1, only the default alterability coefficients (0 for a benchmark and 1 for a sub-annual series) are valid. The specification of alterability coefficients variables is therefore not allowed. If such variables are specified, the procedure ignores them and displays a warning message in the log.
- Alterability coefficients only come into play after the initial sub-annual series has been corrected for the bias (when applicable). For example, this means that specifying an alterability coefficient of 0 for a given sub-annual series data point *will not* result in an unchanged value after benchmarking if bias correction is used.
- Two-level names such as *libref.SAS-data-set* can be used to specify input data sets or to make output data sets permanent.
- If BY variables are specified, they will appear in all the output data sets.

Examples

```
/* Example 1: Simple case with a single quarterly series to benchmark to annual values */

/* Sub-annual (quarterly) series */
data mySeries;
input year
      period
      value;
datalines;
1998 1 1.9
1998 2 2.4
1998 3 3.1
1998 4 2.2
1999 1 2.0
1999 2 2.6
1999 3 3.4
1999 4 2.4
2000 1 2.3
;

/* Annual benchmarks */
data myBenchmarks;
input startyear
      startperiod
      endyear
      endperiod
      value;
datalines;
1998 1 1998 4 10.3
1999 1 1999 4 10.2
;
run;

/* Benchmarking using...
   - recommended RHO value for quarterly series (RHO=0.729)
   - proportional model (LAMBDA=1)
   - bias-corrected sub-annual series with the bias estimated by the procedure
     (BIASOPTION=3)
*/
proc benchmarking
  benchmarks=myBenchmarks
  series=mySeries
  outbenchmarks=outBenchmarks
  outseries=outSeries
  outgraphtable=outGraphTable
  rho=0.729
  lambda=1
  biasoption=3;
run;

proc print data=outseries;
run;

/* Example 2: Two quarterly series to benchmark to annual values,
              with by groups and user-defined alterability coefficients */

/* Sub-annual (quarterly) series where 1999 quarters 1 and 2 for BY group A
   of series van_sales_quarterly are non-alterable (see variable alt_van) */
```

```

data mySeries;
format group $1.;
input group      /* BY group id */
      year
      period
      car_sales_quarterly
      van_sales_quarterly
      alt_van @@;  /* Alterability coefficients for van_sales_quarterly */
datalines;
A 1998 1 1851 1900 1      A 1998 2 2436 2200 1
A 1998 3 3115 3000 1      A 1998 4 2205 2000 1
A 1999 1 1987 1900 0      A 1999 2 2635 2500 0
A 1999 3 3435 3800 1      A 1999 4 2361 2500 1
A 2000 1 2183 2100 1      A 2000 2 2822 3100 1
A 2000 3 3664 3650 1      A 2000 4 2550 2950 1
A 2001 1 2342 3300 1      A 2001 2 3001 4000 1
A 2001 3 3779 3290 1      A 2001 4 2538 2600 1
A 2002 1 2363 2010 1      A 2002 2 3090 3600 1
A 2002 3 3807 3500 1      A 2002 4 2631 2100 1
A 2003 1 2601 2050 1      A 2003 2 3063 3500 1
A 2003 3 3961 4290 1      A 2003 4 2774 2800 1
A 2004 1 2476 2770 1      A 2004 2 3083 3080 1
A 2004 3 3864 3100 1      A 2004 4 2773 2800 1
A 2005 1 2489 3100 1      A 2005 2 3082 2860 1
B 1998 1 1851 1900 1      B 1998 2 2436 2200 1
B 1998 3 3115 3000 1      B 1998 4 2205 2000 1
B 1999 1 1987 1900 1      B 1999 2 2635 2500 1
B 1999 3 3435 3800 1      B 1999 4 2361 2500 1
B 2000 1 2183 2100 1      B 2000 2 2822 3100 1
B 2000 3 3664 3650 1      B 2000 4 2550 2950 1
B 2001 1 2342 3300 1      B 2001 2 3001 4000 1
B 2001 3 3779 3290 1      B 2001 4 2538 2600 1
B 2002 1 2363 2010 1      B 2002 2 3090 3600 1
B 2002 3 3807 3500 1      B 2002 4 2631 2100 1
B 2003 1 2601 2050 1      B 2003 2 3063 3500 1
B 2003 3 3961 4290 1      B 2003 4 2774 2800 1
B 2004 1 2476 2770 1      B 2004 2 3083 3080 1
B 2004 3 3864 3100 1      B 2004 4 2773 2800 1
B 2005 1 2489 3100 1      B 2005 2 3082 2860 1
;
/* Annual benchmarks */
data myBenchmarks;
format group $1.;
input group      /* BY group id */
      startYear
      startPeriod
      endYear
      endPeriod
      car_sales_annual
      van_sales_annual @@;
datalines;
A 1998 1 1998 4 10324 12000      A 1999 1 1999 4 10200 10400
A 2000 1 2000 4 10582 11550      A 2001 1 2001 4 11097 11400
A 2002 1 2002 4 11582 14500      A 2003 1 2003 4 11092 16000
B 1998 1 1998 4 10324 12000      B 1999 1 1999 4 10200 10400
B 2000 1 2000 4 10582 11550      B 2001 1 2001 4 11097 11400
B 2002 1 2002 4 11582 14500      B 2003 1 2003 4 11092 16000
;

/* Benchmarking...
- using recommended RHO value for quarterly series (RHO=0.729)
- with proportional model (LAMBDA=1)
- without bias correction (BIASOPTION=1 and option BIAS= not specified)

```



```

*/
proc benchmarking
  benchmarks=myBenchmarks
  series=mySeries
  outbenchmarks=outBenchmarks
  outseries=outSeries
  outgraphtable=outGraphTable
  rho=0.729
  lambda=1
  biasoption=1;

  var car_sales_quarterly van_sales_quarterly/alt_van;
  with car_sales_annual van_sales_annual;
  by group;
run;

```

Notes

This document is a guide for the use of the BENCHMARKING procedure. PROC BENCHMARKING is part of Statistics Canada's G-Series software formerly known as Forillon.

For more information, please contact the G-Series support team using the [G-Series](#) e-mail address (G-Series@statcan.gc.ca) or consult the web site at [G-Series](#) (Statistics Canada intranet only).

References

Bloem, A. M., R. J. Dippelsman, and N. Ø. Mæhle (2001). **Quarterly National Accounts Manual, Concepts, Data Sources and Compilation**, International Monetary Fund, Washington DC.

Dagum, E. B. and P. Cholette (2006). **Benchmarking, Temporal Distribution and Reconciliation Methods of Time Series**, Springer-Verlag, New York, Lecture Notes in Statistics, Vol. 186

Fortier, S. and B. Quenneville (2007). "Theory and Application of Benchmarking in Business Surveys", **ICES III** proceedings, Introductory Overview Lecture at the International Conference on Establishment Surveys III, June 2007.

Latendresse, E., M. Djona and S. Fortier (2007). "Benchmarking Sub-Annual Series to Annual Totals – From Concepts to SAS® Procedure and SAS® Enterprise Guide® Custom Task", **SAS Global Forum** proceedings, April 2007.

Quenneville, B., S. Fortier, Z.-G. Chen and E. Latendresse (2006). "Recent Developments in Benchmarking to Annual Totals in X-12-ARIMA and at Statistics Canada", proceedings of the **2006 Eurostat Conference on Seasonality, Seasonal Adjustment and Their Implications for Short-Term Analysis and Forecasting**, Luxembourg, May 2006.

The TSRAKING Procedure

Overview

This procedure will restore cross-sectional aggregation constraints in a system of time series. The aggregation constraints may come from a 1 or 2-dimensional table. Optionally, temporal constraints can also be preserved.

Procedure Syntax

PROC TSRAKING <option(s)>;

ID *variable(s)*;

To do this	Use this statement
Identify the variables of the input data set to be transferred to the output data set	ID

PROC TSRAKING Statement

PROC TSRAKING <option(s)>;

To do this	Use this option
Specify the alterability coefficients data set	ALTERABILITY=
Specify the default alterability coefficient for the component series annual totals (temporal constraints)	ALTERANNUAL=
Specify the default alterability coefficient for the SERIES variables (component series)	ALTERSERIES=
Specify the default alterability coefficient for the TOTAL1 variables (1 st dimension control totals)	ALTERTOTAL1=
Specify the default alterability coefficient for the TOTAL2 variables (2 nd dimension control totals)	ALTERTOTAL2=
Specify the input data set	DATA=
Specify the metadata data set	METADATA=
Specify the output data set	OUT=
Specify the tolerance level for the ultimate test in percentage or absolute value	TOLERANCEPERCENT= TOLERANCEVALUE=
Specify the threshold for identification of negative values.	TOLNEGRESULT=
Display intermediate results and information about CPU time in the log	VERBOSE
Display a warning message in the log when a negative value is found (see option TOLNEGRESULT=)	WARNNEGRESULT NOWARNNEGRESULT

Options

ALTERABILITY=SAS-data-set

specifies the SAS data set that contains the alterability coefficients variables. Any alterability coefficients variable must correspond to a component series or a control total, that is, a variable with the same name must be present in the DATA= input data set. If the optional ALTERABILITY= data set is used, the values of these alterability coefficients will override the values specified using the ALTERxxx options (see below). When the DATA= data set contains several observations and the ALTERABILITY= data set contains only one, the alterability coefficients are used (repeated) for all observations of the DATA= data set. Alternatively, the ALTERABILITY= data set may contain as many observations as the DATA= data set.

ALTERANNUAL=positive real number (including 0)

specifies the default alterability coefficient for the component series temporal constraints (annual totals). It is optional and has a default value of 0.0 (binding totals). It will apply to temporal constraints for which an alterability coefficient has not already been specified using variable ALTERANNUAL of the METADATA= data set.

ALTERSERIES=positive real number (including 0)

specifies the default alterability coefficient for the component series. It is optional and has a default value of 1.0. It will apply to component series for which an alterability coefficient has not already been specified using the ALTERABILITY= data set.

ALCERTOTAL1=positive real number (including 0)

specifies the default alterability coefficient for the 1st dimension control totals. It is optional and has a default value of 0.0 (binding totals). It will apply to control totals for which an alterability coefficient has not already been specified using the ALTERABILITY= data set.

ALCERTOTAL2=positive real number (including 0)

specifies the default alterability coefficient for the 2nd dimension control totals. It is optional and has a default value of 0.0 (binding totals). It will apply to control totals for which an alterability coefficient has not already been specified using the ALTERABILITY= data set.

DATA=SAS-data-set

specifies the SAS data set that contains the variables representing the time series system, that is the component series and cross-sectional control totals to be raked. If more than one observation is provided, the sum of the provided values will also be preserved as part of implicit temporal constraints. The DATA= data set is mandatory.

METADATA=SAS-data-set

specifies the metadata SAS data set that describes the aggregation constraints (additivity rules) used by the procedure. The METADATA= data set is mandatory. Two character variables must be in this data set: SERIES and TOTAL1. Two variables are optional: TOTAL2 (character) and ALTERANNUAL (numeric). The values of variable SERIES represent the variable names of the component series in the

DATA= input data set. Similarly, the values of variables TOTAL1 and TOTAL2 represent the variable names of the 1st and 2nd dimensions control totals in the DATA= input data set. Variable ALTERANNUAL contains the alterability coefficient for the temporal constraint associated to the component series.

OUT=SAS-data-set

specifies the output SAS data set that contains all the raked component series, raked cross-sectional control totals and any variables specified in the ID statement. It is optional. If it is not specified, PROC TSRAKING will create it by using the DATA n naming convention.

TOLERANCEPERCENT= | **TOLERANCEVALUE=**
TOLP=*positive real number (including 0)* | **TOLV=***positive real number (including 0)*

specifies the tolerance, in percentage or absolute value, to be used when performing the ultimate test in the case of binding totals (alterability coefficient = 0.0 for the control totals). The test compares the binding control totals with the ones calculated from raked component series. This number is optional.

Default:	TOLERANCEVALUE=1E-3.
Example:	To set a tolerance of 1%, specify TOLERANCEPERCENT=0.01 To set a tolerance of 10, specify TOLERANCEVALUE=10

TOLNEGRESULT=
TOLN=*strictly negative real number (excluding 0)*

specifies the threshold for identification of negative values. A value is considered negative when it is smaller than this threshold. This number is optional. The default value for this option is -1E-3.

VERBOSE

displays intermediate results and information about CPU time in the log.

WARNNEGRESULT | **NOWARNNEGRESULT**

specifies whether PROC TSRAKING displays a warning message in the log when a negative value created by the procedure is smaller than the threshold specified by the TOLNEGRESULT= option. The default option is WARNNEGRESULT.

ID Statement

ID *variable-1 ... variable-n;*

Use the ID statement to name additional variables to be transferred from the DATA= input data set to the OUT= output data set. By default, the output data set only contains the variables described in the metadata (see option METADATA=). Variables listed in the ID statement must be present in the DATA= input data set.

Details

- With the exception of the variables listed in the ID statement, missing values in the DATA= input data set will stop the procedure.
 - Missing and/or negative values in the ALTERABILITY= input data set will stop the procedure.
 - Two-level names such as *libref.SAS-data-set* can be used to specify input data sets or to make the output data set permanent.
-

Examples

```
/* EXAMPLE 1 */
```

```
/* In this 1-dimensional example, the values of "cars" and "vans" must sum up to the value of "total" */
```

```
data myMetadata;  
input series $5.  
      Total1 $5.;  
datalines;  
cars total  
vans total  
;
```

```
data myData;  
input cars  
      vans  
      total;  
datalines;  
25 5 40  
;
```

```
proc tsraking  
  metadata=myMetadata  
  data=myData  
  out=outData;  
run;
```

```
/* EXAMPLE 2 */
```

```
/* In this 2-dimensional example, we have estimates of "cars" and "vans" sales for the "alb", "sask" and "man" regions. The sum of the 3 regions for "cars" must be equal to its control total (cars_alb + cars_sask + cars_man = cars_total). The sum of the 3 regions for "vans" must be equal to its control total (vans_alb + vans_sask + vans_man = vans_total). Also, in each of the regions, the sum of "cars" and "vans" must be equal to the regional control total (cars_alb + vans_alb = alb_total; cars_sask + vans_sask = sask_total and cars_man + vans_man = man_total). Finally, "vans" sales in region "sask" (vans_sask) are non-alterable (alterability coefficient = 0) */
```

```
data myMetadata;  
input series $9.  
      Total1 $11.  
      Total2 $11.;  
datalines;  
cars_alb cars_total alb_total  
cars_sask cars_total sask_total  
cars_man cars_total man_total
```

```

vans_alb  vans_total alb_total
vans_sask vans_total sask_total
vans_man  vans_total man_total
;

data myData;
input cars_alb cars_sask cars_man vans_alb vans_sask vans_man
      alb_total sask_total man_total cars_total vans_total;
datalines;
12 14 13 20 20 24 30 31 32 40 53
;

data myalter;
input cars_alb cars_sask cars_man vans_alb vans_sask vans_man
      alb_total sask_total man_total cars_total vans_total;
datalines;
1 1 1 1 0 1 0 0 0 0 0
;

proc tsraking
  metadata=myMetadata
  data=myData
  alterability=myAlter
  out=outData;
run;

/* EXAMPLE 3 */

/* In this 2-dimensional example with temporal constraints, we have the constraints  $A_1 + A_2 + A_3 = A$ ;  $B_1 + B_2 + B_3 = B$ ;  $A_1 + B_1 = \_1$ ;  $A_2 + B_2 = \_2$  and  $A_3 + B_3 = \_3$ . We have 4 (quarterly) observations of each value and we want to preserve annual totals. For each variable, the sum of the 4 quarterly values must stay the same after the reconciliation step. */

data mymetadata;
input series $3.
      total1 $3.
      total2 $3.;
datalines;
A_1 _1 A
B_1 _1 B
A_2 _2 A
B_2 _2 B
A_3 _3 A
B_3 _3 B
;

data mydata;
input A_1 A_2 A_3 B_1 B_2 B_3 A B _1 _2 _3;
datalines;
12 14 13 20 20 24 40 53 30 31 32
10 9 15 21 29 20 25 80 35 35 35
12 8 17 15 20 30 40 59 23 32 44
9 9 14 17 24 23 37 71 28 35 45
;

proc tsraking
  metadata=myMetadata
  data=myData
  out=outData;
run;

```

Notes

This document is a guide for the use of the TSRAKING procedure. PROC TSRAKING is part of Statistics Canada's G-Series software formerly known as Forillon.

For more information, please contact the G-Series support team using the [G-Series](#) e-mail address (G-Series@statcan.gc.ca) or consult the web site at [G-Series](#) (Statistics Canada intranet only).

References

Bérubé, J. and S. Fortier (2009). "PROC TSRAKING: An in-house SAS procedure for balancing time series", **JSM 2009** proceedings, Business and Economic Section. Alexandria, VA: American Statistical Association.

Dagum, E. B. and P. Cholette (2006). **Benchmarking, Temporal Distribution and Reconciliation Methods of Time Series**, Springer-Verlag, New York, Lecture Notes in Statistics, Vol. 186

Fortier, S. and B. Quenneville (2009). "Reconciliation and Balancing of Accounts and Time Series", **JSM 2009** proceedings, Business and Economic Section. Alexandria, VA: American Statistical Association.
