

La procédure **BENCHMARKING**

Aperçu

Cette procédure assure la cohérence entre des séries chronologiques d’une même variable cible mesurée à différents intervalles (ex : infra-annuelle et annuelle). L’étalonnage consiste à imposer le niveau de la série d’étalons tout en minimisant, autant que possible, les révisions au mouvement observé dans la série infra-annuelle. La procédure permet également l’étalonnage non contraignant où la série d’étalons peut aussi être révisée.

La procédure peut aussi être utilisée à d’autres fins reliées à l’étalonnage, telles que la distribution temporelle (désagrégation de la série d’étalons en observations plus fréquentes), la calendarisation (cas particulier de distribution temporelle) et le raccordement (connexion de différents segments de séries chronologiques en une série unique cohérente).

Syntaxe de la procédure

PROC BENCHMARKING <option(s)>;

VAR var1 </alt1> <...varN </altN> >;

WITH var1 </alt1> <...varN </altN> >;

BY variable(s);

Pour faire ceci	Utilisez cet énoncé
Procéder à un traitement séparé pour les observations de chaque groupe BY	BY
Identifier, au besoin, la ou les séries infra-annuelles à étalonner ainsi que les coefficients d’altérabilité associés	VAR
Identifier, au besoin, la ou les séries d’étalons ainsi que les coefficients d’altérabilité associés	WITH

Énoncé **PROC BENCHMARKING**

PROC BENCHMARKING <option(s)>;

Pour faire ceci	Utilisez cette option
Spécifier l’ensemble de données d’entrée qui contient les étalons	BENCHMARKS=
Spécifier une valeur pour le biais	BIAS=
Spécifier l’option pour l’estimation du biais	BIASOPTION=
Spécifier la valeur du paramètre λ (lambda)	LAMBDA=
Spécifier l’ensemble de données de sortie qui contient les étalons utilisés	OUTBENCHMARKS=

par la procédure	
Spécifier l'ensemble de données de sortie qui contient des données supplémentaires utiles à la production de tableaux analytiques et de graphiques	OUTGRAPHTABLE=
Spécifier l'ensemble de données de sortie qui contient la ou les séries infra-annuelles étalonnées	OUTSERIES=
Spécifier la valeur pour le paramètre ρ (rho)	RHO=
Spécifier l'ensemble de données d'entrée qui contient la ou les séries infra-annuelles à étalonner	SERIES=
Spécifier le niveau de tolérance pour le test ultime, en pourcentage ou en valeur absolue	TOLERANCEPERCENT= TOLERANCEVALUE=
Spécifie le seuil pour l'identification des valeurs négatives	TOLNEGRESULT=
Afficher des résultats intermédiaires et le temps de CPU requis dans le journal (SAS log)	VERBOSE
Afficher un message d'avertissement dans le journal (SAS log) lorsqu'une valeur négative est trouvée (voir option TOLNEGRESULT=)	WARNNEGRESULT NOWARNNEGRESULT

Options

BENCHMARKS=*ensemble-de-données-SAS*

spécifie l'ensemble de données d'entrée SAS qui contient les étalons. Cette option est obligatoire. L'ensemble de données doit contenir les variables numériques qui définissent la période de couverture de chaque étalon, soit: STARTYEAR, STARTPERIOD, ENDYEAR et ENDPERIOD. De plus, une variable numérique nommée VALUE (par défaut) contenant la valeur de l'étalon doit obligatoirement se trouver dans le fichier de données. Pour spécifier un autre nom de variable ou plus d'une série d'étalons, utilisez l'énoncé VAR et/ou WITH.

BIAS=*nombre réel*

spécifie la valeur du biais défini par l'utilisateur à utiliser pour la correction de la série infra-annuelle avant l'étalonnage. Le biais est ajouté à la série infra-annuelle avec un modèle additif (lorsque LAMBDA=0.0) et multiplié autrement (lorsque LAMBDA≠0.0). L'option BIAS= n'est pas obligatoire. La valeur par défaut est BIAS=0.0 lorsque LAMBDA=0.0 et BIAS=1.0 autrement. Il est à noter que l'option BIAS= n'est pas utilisée lorsque l'option BIASOPTION=3 (voir ci-dessous).

BIASOPTION=1, 2 ou 3

spécifie l'option pour l'estimation du biais. Cette option est obligatoire.

Si vous spécifiez la valeur suivante pour l'estimation du biais	La procédure
1	N'estimera pas le biais. Le biais utilisé pour corriger la série infra-annuelle sera la valeur spécifiée avec l'option BIAS= ou la valeur par défaut (voir l'option BIAS=).
2	Estimera le biais, imprimera le résultat dans le journal mais ne l'utilisera pas. Le biais utilisé pour corriger la série infra-annuelle sera la valeur spécifiée avec l'option BIAS= ou la valeur par défaut (voir l'option BIAS=).

Estimera le biais, imprimera le résultat dans le journal et appliquera la correction à l'aide de la valeur estimée du biais. Une valeur spécifiée à l'aide de l'option BIAS= sera imprimée dans le journal mais ne sera pas utilisée.

LAMBDA=nombre réel (valeurs suggérées entre -3 et 3)

spécifie la valeur du paramètre d'ajustement du modèle λ . Cette option est obligatoire. Les valeurs fréquemment utilisées sont LAMBDA=0.0 pour un modèle additif et LAMBDA=1.0 pour un modèle proportionnel. On obtient un ajustement au prorata en spécifiant LAMBDA=0.5 et RHO=0.0.

OUTBENCHMARKS=ensemble-de-données-SAS

spécifie l'ensemble de données de sortie SAS qui contiendra les étalons utilisés par la procédure. Si le nom de l'ensemble de données de sortie n'est pas spécifié, alors PROC BENCHMARKING le créera en utilisant les règles d'affectation des noms DATA n . L'ensemble de données contiendra les mêmes variables que l'ensemble de données BENCHMARKS=. Si des variables BY sont spécifiées, elles se retrouveront aussi dans cet ensemble de données.

OUTGRAPHTABLE=ensemble-de-données-SAS

spécifie l'ensemble de données de sortie SAS qui contiendra des données supplémentaires utiles à la production de tableaux analytiques et de graphiques dans le cas d'étalons sans chevauchement. Cette option n'est pas obligatoire et aucun ensemble de données de sortie ne sera créé si elle n'est pas utilisée. Le tableau suivant décrit les variables qui se trouvent dans cet ensemble de données. Si des variables BY sont spécifiées, elles apparaîtront aussi dans cet ensemble de données.

Nom de la variable	Description
ALTBENCHMARKS	Le nom de la variable utilisée pour spécifier les coefficients d'altérabilité de la série d'étalons
ALTBENCHMARKSVALUE	Les valeurs des coefficients d'altérabilité de la série d'étalons
ALTSERIES	Le nom de la variable utilisée pour spécifier les coefficients d'altérabilité de la série infra-annuelle
ALTSERIESVALUE	Les valeurs des coefficients d'altérabilité de la série infra-annuelle
AVGBENCHMARK	La valeur de l'étalon divisée par le nombre de périodes couvertes; valeur manquante s'il n'y a pas d'étalon
AVGBENCHMARKSUBANNUALRATIO	Les ratios ^{**} entre les valeurs des variables AVGBENCHMARK et AVGSUBANNUAL; valeur manquante s'il n'y a pas d'étalon
AVGSUBANNUAL	La moyenne des valeurs de la série infra-annuelle initiale par période de couverture de l'étalon; valeur manquante s'il n'y a pas d'étalon
BENCHMARKED	Les valeurs de la série infra-annuelle étalonnée
BENCHMARKEDSUBANNUALRATIO	Les ratios ^{**} entre les valeurs des variables BENCHMARKED et SUBANNUAL
BIAS	La valeur du BIAIS utilisée (calculée par la procédure ou fournie par l'utilisateur)

DATE	Variable de type caractère qui combine les valeurs des variables YEAR et PERIOD
GROWTHRATEBENCHMARKED	Les taux de croissance ^{**} dans la série infra-annuelle étalonnée
GROWTHRATESUBANNUAL	Les taux de croissance ^{**} dans la série infra-annuelle initiale
LAMBDA	La valeur du paramètre λ (lambda)
M	Les identificateurs de période de couverture des étalons, $m = 1, \dots, M$; valeur manquante s'il n'y a pas d'étalon
PERIOD	Valeurs des périodes
PERIODICITY	Le nombre maximum de périodes à l'intérieur d'une année (ex : la périodicité est 4 pour une série trimestrielle)
RHO	La valeur du paramètre ρ (rho)
SUBANNUAL	Les valeurs de la série infra-annuelle initiale
SUBANNUALCORRECTED	Les valeurs de la série infra-annuelle corrigée (données rectifiées avec le biais)
T	Les identificateurs des périodes infra-annuelles, $t = 1, \dots, T$
VARBENCHMARKS	Le nom de la variable utilisée pour spécifier la série d'étalons
VARSERIES	Le nom de la variable utilisée pour spécifier la série infra-annuelle
YEAR	Valeurs des années

^{**} Les termes « ratios » et « taux de croissance » réfèrent ici à une terminologie couramment utilisée dans le domaine des séries chronologiques et signifient en fait « différences » et « premières différences » respectivement lorsque le modèle additif est utilisé (LAMBDA=0.0).

OUTSERIES=*ensemble-de-données-SAS*

spécifie l'ensemble de données de sortie SAS qui contiendra la ou les séries infra-annuelles étalonnées. Si le nom de l'ensemble de données de sortie n'est pas spécifié, alors PROC BENCHMARKING le créera en utilisant les règles d'affectation des noms DATA n . L'ensemble de données contiendra les mêmes variables que l'ensemble de données SERIES=. Si des variables BY sont spécifiées, elles se retrouveront aussi dans cet ensemble de données.

RHO=*nombre réel entre 0 et 1 (inclusif)*

spécifie la valeur du paramètre autorégressif ρ dans l'intervalle [0,1]. Cette option est obligatoire.

SERIES=*ensemble-de-données-SAS*

spécifie l'ensemble de données d'entrée SAS qui contient la série infra-annuelle à étalonner. Cette option est obligatoire. L'ensemble de données doit contenir les variables numériques YEAR et PERIOD. De plus, une variable numérique nommée VALUE (par défaut) contenant la valeur de la série infra-annuelle à étalonner doit obligatoirement se trouver dans le fichier de données. Pour spécifier un autre nom de variable ou plus d'une série infra-annuelle, utilisez l'énoncé VAR.

TOLERANCEPERCENT= | **TOLERANCEVALUE=**
TOLP=*nombre réel positif (incluant 0)* | **TOLV=***nombre réel positif (incluant 0)*

spécifie le niveau de tolérance, en pourcentage ou en valeur absolue, à utiliser pour le test ultime dans le cas d'étalons contraignants (coefficient d'altérabilité = 0.0 pour les étalons). Ce test compare les étalons fournis à l'entrée avec ceux calculés à partir de la série infra-annuelle étalonnée. Ce nombre est optionnel.

Par défaut :	TOLERANCEVALUE=1E-3.
Exemple :	Pour une tolérance de 1%, spécifiez TOLERANCEPERCENT =0.01 Pour une tolérance de 10, spécifiez TOLERANCEVALUE=10

TOLNEGRESULT=
TOLN=*nombre réel négatif (excluant 0)*

spécifie le seuil pour l'identification des valeurs négatives. Une valeur est considérée négative lorsqu'elle est inférieure à ce seuil. Ce nombre est optionnel. La valeur par défaut est -1E-3.

VERBOSE

affiche des résultats intermédiaires et le temps de CPU requis dans le journal (SAS log). Avec cette option, les matrices intermédiaires seront imprimées dans le journal ainsi que le temps de CPU utilisé pour inverser les matrices, effectuer les multiplications de matrices, etc.

WARNNEGRESULT | NOWARNNEGRESULT

spécifie si la procédure doit afficher un message d'avertissement dans le journal (SAS log) lorsqu'une valeur négative créée par la procédure est inférieure au seuil tel que spécifié par l'option TOLNEGRESULT=. L'option par défaut est WARNNEGRESULT.

Énoncé VAR

VAR *var1* </alt1> <...*varN* </altN> >;

Arguments requis

var1...varN

spécifie la ou les variables qui contiennent les valeurs des séries infra-annuelles à étalonner. Les variables doivent être de type numérique. L'énoncé VAR est optionnel. Si cet énoncé n'est pas spécifié, la procédure cherchera la variable VALUE dans le fichier d'entrée SERIES=.

alt1...altN

spécifie au besoin la ou les variables qui contiennent les coefficients d'altérabilité applicables aux séries infra-annuelles. Les variables de coefficients d'altérabilité doivent être de type numérique. Dans le cas où une variable de coefficients d'altérabilité n'est pas spécifiée pour une série infra-annuelle donnée, la procédure utilisera la valeur par défaut d'un coefficient d'altérabilité pour une série infra-annuelle, soit 1.0. Les coefficients d'altérabilité entrent en jeu seulement après que la série infra-annuelle initiale ait été

corrigée pour le biais (le cas échéant). Par exemple, ça signifie que la spécification d'un coefficient d'altérabilité de 0.0 pour période donnée de la série infra-annuelle *ne* résultera *pas* en une valeur inchangée après étalonnage s'il y a correction pour le biais (se référer aux options BIAS= et BIASOPTION=). La spécification de coefficients d'altérabilité n'est pas permise lorsque RHO=1.

Énoncé WITH

WITH var1 </alt1> <...varN </altN> >;

Arguments requis

var1...varN

spécifie la ou les variables qui contiennent les valeurs des séries d'étalons. Les variables doivent être de type numérique. L'énoncé VAR est optionnel.. Si cet énoncé n'est pas spécifié, la procédure cherchera la variable VALUE dans le fichier d'entrée BENCHMARKS=.

alt1...altN

spécifie au besoin la ou les variables qui contiennent les coefficients d'altérabilité applicables aux étalons. Les variables de coefficients d'altérabilité doivent être de type numérique. Dans le cas où une variable de coefficients d'altérabilité n'est pas spécifiée pour une série d'étalons donnée, la procédure utilisera la valeur par défaut d'un coefficient d'altérabilité pour un étalon, soit 0.0 (étalons contraignants). La spécification de coefficients d'altérabilité n'est pas permise lorsque RHO=1.

Énoncé BY

BY variable-1 <... variable-n>;

Arguments requis

variable(s)

spécifie la ou les variables que la procédure utilise pour former les groupes BY. Les variables BY doivent être présentes dans les deux fichiers d'entrée (BENCHMARKS= et SERIES=) et elles apparaîtront dans les trois fichiers de sortie (OUTBENCHMARKS=, OUTGRAPHTABLE= et OUTSERIES=). Plus d'une variable peut être spécifiée. Cet énoncé est optionnel. Les variables peuvent être de type caractère ou numérique.

Détails

- Si une valeur manquante apparaît dans une des variables du fichier d'entrée BENCHMARKS= (autre que les variables BY), les observations avec valeurs manquantes sont retirées, un message d'avertissement est affiché dans le journal et la procédure poursuit son exécution.

- Si une valeur manquante apparaît dans les variable YEAR et/ou PERIOD du fichier d'entrée SERIES= et que des variables BY sont spécifiées, le groupe BY correspondant est retiré, un message d'avertissement est affiché dans le journal et la procédure traite les autres groupes BY. Si aucune variable BY n'est spécifiée, aucun traitement n'est effectué et un message d'avertissement apparaît dans le journal.
- Si une valeur manquante apparaît dans une variable désignant une série infra-annuelle dans le fichier d'entrée SERIES= (voir l'énoncé VAR) et que des variables BY sont spécifiées, le groupe BY correspondant est retiré, un message d'avertissement est affiché dans le journal et la procédure traite les autres groupes BY. Si aucune variable BY n'est spécifiée, aucun traitement n'est effectué sur la série infra-annuelle en question, un message d'avertissement apparaît dans le journal et la procédure poursuit son traitement avec la série infra-annuelle suivante (le cas échéant).
- La procédure ne permet pas l'étalonnage proportionnel ($LAMBDA \neq 0$) d'une série indicatrice qui est nulle sur toute la période de couverture d'un étalon non-nul. Une telle situation génère un message d'erreur dans le journal (SAS log).
- Avec une valeur de $RHO=1$, seuls les coefficients d'altérabilité par défaut (0 pour un étalon et 1 pour une série infra-annuelle) sont valides. La spécification de variables contenant les coefficients d'altérabilité n'est donc pas permise. Si tel est le cas, la procédure les ignore et affiche un message d'avertissement dans le journal (SAS log).
- Les coefficients d'altérabilité entrent en jeu seulement après que la série infra-annuelle initiale ait été corrigée pour le biais (le cas échéant). Par exemple, ça signifie que la spécification d'un coefficient d'altérabilité de 0 pour période donnée de la série infra-annuelle *ne* résultera *pas* en une valeur inchangée après étalonnage s'il y a correction pour le biais.
- Des noms à deux niveaux tels que *libref.ensemble-de-données-SAS* peuvent être utilisés pour les fichiers d'entrée et pour rendre permanent les fichiers de sortie.
- Si des variables BY sont spécifiées, celles-ci apparaîtront dans tous les fichiers de sortie.

Exemples

```
/* Exemple 1: Cas simple avec une seule série trimestrielle à étalonner à des valeurs
   annuelles */
```

```
/* Série infra-annuelle (trimestrielle) */
```

```
data mySeries;
```

```
input year
       period
       value;
```

```
datalines;
```

```
1998 1 1.9
1998 2 2.4
1998 3 3.1
1998 4 2.2
1999 1 2.0
1999 2 2.6
1999 3 3.4
1999 4 2.4
2000 1 2.3
;
```

```
/* Étalons annuels */
```

```
data myBenchmarks;
```

```
input startyear
       startperiod
       endyear
```

```

        endperiod
        value;
datalines;
1998 1 1998 4 10.3
1999 1 1999 4 10.2
;
run;

/* Étalonnage en utilisant...
- la valeur recommandée de RHO pour des séries trimestrielles (RHO=0.729)
- un modèle proportionnel (LAMBDA=1)
- une série infra-annuelle corrigée pour le biais estimé par la procédure
  (BIASOPTION=3)
*/
proc benchmarking
    benchmarks=myBenchmarks
    series=mySeries
    outbenchmarks=outBenchmarks
    outseries=outSeries
    outgraphtable=outGraphTable
    rho=0.729
    lambda=1
    biasoption=3;
run;

proc print data=outseries;
run;

/* Exemple 2: Deux séries trimestrielles à étalonner à des totaux annuels, avec des
groupes BY et des coefficients d'altérabilité définis par l'utilisateur.
*/

/* Séries infra-annuelles (trimestrielles) où les 1er et 2e trimestres de 1999 pour le
groupe BY « A » de la série van_sales_quarterly sont non-altérables (voir la variable
alt_van) */
data mySeries;
format group $1.;
input group /* Identificateur des groupes BY */
    year
    period
    car_sales_quarterly
    van_sales_quarterly
    alt_van @@; /* Coefficients d'altérabilité pour van_sales_quarterly */
datalines;
A 1998 1 1851 1900 1    A 1998 2 2436 2200 1
A 1998 3 3115 3000 1    A 1998 4 2205 2000 1
A 1999 1 1987 1900 0    A 1999 2 2635 2500 0
A 1999 3 3435 3800 1    A 1999 4 2361 2500 1
A 2000 1 2183 2100 1    A 2000 2 2822 3100 1
A 2000 3 3664 3650 1    A 2000 4 2550 2950 1
A 2001 1 2342 3300 1    A 2001 2 3001 4000 1
A 2001 3 3779 3290 1    A 2001 4 2538 2600 1
A 2002 1 2363 2010 1    A 2002 2 3090 3600 1
A 2002 3 3807 3500 1    A 2002 4 2631 2100 1
A 2003 1 2601 2050 1    A 2003 2 3063 3500 1
A 2003 3 3961 4290 1    A 2003 4 2774 2800 1
A 2004 1 2476 2770 1    A 2004 2 3083 3080 1
A 2004 3 3864 3100 1    A 2004 4 2773 2800 1
A 2005 1 2489 3100 1    A 2005 2 3082 2860 1
B 1998 1 1851 1900 1    B 1998 2 2436 2200 1
B 1998 3 3115 3000 1    B 1998 4 2205 2000 1
B 1999 1 1987 1900 1    B 1999 2 2635 2500 1

```



```

B 1999 3 3435 3800 1      B 1999 4 2361 2500 1
B 2000 1 2183 2100 1      B 2000 2 2822 3100 1
B 2000 3 3664 3650 1      B 2000 4 2550 2950 1
B 2001 1 2342 3300 1      B 2001 2 3001 4000 1
B 2001 3 3779 3290 1      B 2001 4 2538 2600 1
B 2002 1 2363 2010 1      B 2002 2 3090 3600 1
B 2002 3 3807 3500 1      B 2002 4 2631 2100 1
B 2003 1 2601 2050 1      B 2003 2 3063 3500 1
B 2003 3 3961 4290 1      B 2003 4 2774 2800 1
B 2004 1 2476 2770 1      B 2004 2 3083 3080 1
B 2004 3 3864 3100 1      B 2004 4 2773 2800 1
B 2005 1 2489 3100 1      B 2005 2 3082 2860 1
;
/* Étalons annuels */
data myBenchmarks;
format group $1.;
input group      /* Identificateur des groupes BY */
      startYear
      startPeriod
      endYear
      endPeriod
      car_sales_annual
      van_sales_annual @@;
datalines;
A 1998 1 1998 4 10324 12000      A 1999 1 1999 4 10200 10400
A 2000 1 2000 4 10582 11550      A 2001 1 2001 4 11097 11400
A 2002 1 2002 4 11582 14500      A 2003 1 2003 4 11092 16000
B 1998 1 1998 4 10324 12000      B 1999 1 1999 4 10200 10400
B 2000 1 2000 4 10582 11550      B 2001 1 2001 4 11097 11400
B 2002 1 2002 4 11582 14500      B 2003 1 2003 4 11092 16000
;

/* Étalonnage en utilisant...
- la valeur recommandée de RHO pour des séries trimestrielles (RHO=0.729)
- un modèle proportionnel (LAMBDA=1)
- une série infra-annuelle non corrigée pour le biais (BIASOPTION=1 et option
  BIAS= non spécifiée)
*/
proc benchmarking
  benchmarks=myBenchmarks
  series=mySeries
  outbenchmarks=outBenchmarks
  outseries=outSeries
  outgraphtable=outGraphTable
  rho=0.729
  lambda=1
  biasoption=1;

  var car_sales_quarterly van_sales_quarterly/alt_van;
  with car_sales_annual van_sales_annual;
  by group;
run;

```

Notes

Ce document est un guide pour l'utilisation de la procédure BENCHMARKING. PROC BENCHMARKING fait partie du logiciel G-Series de Statistique Canada, précédemment appelé Forillon.

Pour plus de renseignements, veuillez vous adresser à l'équipe de soutien de G-Series à l'aide de l'adresse courriel de [G-Series](mailto:G-Series@statcan.gc.ca) (G-Series@statcan.gc.ca) ou consultez le site web de [G-Series](#) (Intranet de Statistique Canada seulement).

Bibliographie

Bloem, A. M., R. J. Dippelsman, et N. Ø. Mæhle (2001). **Quarterly National Accounts Manual, Concepts, Data Sources and Compilation**, International Monetary Fund, Washington DC.

Dagum, E. B. et P. Cholette (2006). **Benchmarking, Temporal Distribution and Reconciliation Methods of Time Series**, Springer-Verlag, New York, Lecture Notes in Statistics, Vol. 186

Fortier, S. et B. Quenneville (2007). « Theory and Application of Benchmarking in Business Surveys », recueil de la conférence **ICES III**, Introductory Overview Lecture at the International Conference on Establishment Surveys III, juin 2007.

Latendresse, E., M. Djona et S. Fortier (2007). « Benchmarking Sub-Annual Series to Annual Totals – From Concepts to SAS® Procedure and SAS® Enterprise Guide® Custom Task », recueil de la conférence **SAS Global Forum**, avril 2007.

Quenneville, B., S. Fortier, Z.-G. Chen et E. Latendresse (2006). « Recent Developments in Benchmarking to Annual Totals in X-12-ARIMA and at Statistics Canada », recueil de la conférence **2006 Eurostat Conference on Seasonality, Seasonal Adjustment and Their Implications for Short-Term Analysis and Forecasting**, Luxembourg, mai 2006.

La procédure TSRAKING

Aperçu

Cette procédure restaure des contraintes d'agrégation transversales dans un système de séries chronologiques. Les contraintes d'agrégation peuvent provenir d'une table à 1 ou 2 dimensions. Optionnellement, des contraintes temporelles peuvent aussi être préservées.

Syntaxe de la procédure

PROC TSRAKING <option(s)>;

ID *variable(s)*;

Pour faire ceci	Utilisez cet énoncé
Identifier les variables du fichier d'entrée qui seront transférées dans le fichier de sortie.	ID

Énoncé PROC TSRAKING

PROC TSRAKING <option(s)>;

Pour faire ceci	Utilisez cette option
Spécifier l'ensemble de données contenant les coefficients d'altérabilité	ALTERABILITY=
Spécifier la valeur par défaut du coefficient d'altérabilité pour les totaux annuels des séries composantes (contraintes temporelles)	ALTERANNUAL=
Spécifier la valeur par défaut du coefficient d'altérabilité pour les variables SERIES (séries composantes)	ALTERSERIES=
Spécifier la valeur par défaut du coefficient d'altérabilité pour les variables TOTAL1 (totaux de contrôle de la 1 ^{ère} dimension)	ALTERTOTAL1=
Spécifier la valeur par défaut du coefficient d'altérabilité pour les variables TOTAL2 (totaux de contrôle de la 2 ^e dimension)	ALTERTOTAL2=
Spécifier l'ensemble de données d'entrée	DATA=
Spécifier l'ensemble de données contenant les métadonnées	METADATA=
Spécifier l'ensemble de données de sortie	OUT=
Spécifier le niveau de tolérance souhaité pour le test ultime, en pourcentage ou en valeur absolue	TOLERANCEPERCENT= TOLERANCEVALUE=
Spécifier le seuil pour l'identification des valeurs négatives	TOLNEGRESULT=
Afficher des résultats intermédiaires et le temps de CPU requis dans le journal (SAS log)	VERBOSE
Afficher un message d'avertissement dans le journal (SAS log) lorsqu'une valeur négative est trouvée (voir option TOLNEGRESULT=)	WARNNEGRESULT NOWARNNEGRESULT

Options

ALTERABILITY=*ensemble-de-données-SAS*

spécifie l'ensemble de données SAS qui contient les variables de coefficients d'altérabilité. Toute variable de coefficient d'altérabilité doit correspondre à une série composante ou un total de contrôle, c'est-à-dire qu'une variable avec le même nom doit être présente dans l'ensemble de données d'entrée DATA=. L'option ALTERABILITY= n'est pas obligatoire. Si elle est utilisée, les coefficients d'altérabilité de cet ensemble de données remplaceront ceux spécifiés avec les options ALTERxxx (voir plus bas). Lorsque l'ensemble de données DATA= contient plusieurs observations et que l'ensemble de données ALTERABILITY= n'en contient qu'une seule, les coefficients d'altérabilité qu'il contient sont utilisés (répétés) pour toutes les observations de l'ensemble de données DATA=. Alternativement, l'ensemble de données ALTERABILITY= peut ne contenir le même nombre d'observations que l'ensemble de données DATA=.

ALTERANNUAL=*nombre réel positif (incluant 0)*

spécifie la valeur par défaut du coefficient d'altérabilité pour les contraintes temporelles (totaux annuels) des séries composantes. Ce nombre est optionnel et la valeur par défaut est 0.0 (totaux contraignants). Cette valeur s'appliquera aux contraintes temporelles pour lesquelles un coefficient d'altérabilité n'a pas déjà été spécifié à l'aide de la variable ALTERANNUAL de l'ensemble de données METADATA=.

ALTERSERIES=*nombre réel positif (incluant 0)*

spécifie la valeur par défaut du coefficient d'altérabilité pour les séries composantes. Ce nombre est optionnel et la valeur par défaut est 1.0. Cette valeur s'appliquera aux séries composantes pour lesquelles un coefficient d'altérabilité n'a pas déjà été spécifié à l'aide de l'ensemble de données ALTERABILITY =.

ALTERTOTAL1=*nombre réel positif (incluant 0)*

spécifie la valeur par défaut du coefficient d'altérabilité pour les totaux de contrôle de la 1^{ère} dimension. Ce nombre est optionnel et la valeur par défaut est 0.0 (totaux contraignants). Cette valeur s'appliquera aux totaux de contrôle pour lesquels un coefficient d'altérabilité n'a pas déjà été spécifié à l'aide de l'ensemble de données ALTERABILITY =.

ALTERTOTAL2=*nombre réel positif (incluant 0)*

spécifie la valeur par défaut du coefficient d'altérabilité pour les totaux de contrôle de la 2^e dimension. Ce nombre est optionnel et la valeur par défaut est 0.0 (totaux contraignants). Cette valeur s'appliquera aux totaux de contrôle pour lesquels un coefficient d'altérabilité n'a pas déjà été spécifié à l'aide de l'ensemble de données ALTERABILITY =.

DATA=*ensemble-de-données-SAS*

spécifie l'ensemble de données SAS qui contient les variables représentant le système de séries chronologiques, c'est-à-dire les séries composantes et totaux de contrôle transversaux à réconcilier. Lorsque l'ensemble de données DATA= contient plusieurs observations, la somme des valeurs de chaque variable est également préservée, définissant des contraintes temporelles implicites. L'option DATA= est obligatoire.

METADATA=ensemble-de-données-SAS

spécifie l'ensemble de données SAS qui contient les métadonnées décrivant les contraintes d'agrégation (règles d'additivité) utilisées par la procédure. L'option METADATA= est obligatoire. L'ensemble de données doit contenir deux variables de type caractère : SERIES et TOTAL1. Deux autres variables sont optionnelles : TOTAL2 (de type caractère) et ALTERANNUAL (de type numérique). Les valeurs de la variable SERIES représentent les noms de variable des séries composantes dans l'ensemble de données d'entrée DATA=. Pareillement, les valeurs des variables TOTAL1 et TOTAL2 représentent les noms de variable des totaux de contrôle des 1^{ère} et 2^e dimensions respectivement dans l'ensemble de données d'entrée DATA=. La variable ALTERANNUAL quant à elle, contient le coefficient d'altérabilité pour la contrainte temporelle (total annuel) associée à la série composante.

OUT=ensemble-de-données-SAS

spécifie l'ensemble de données de sortie SAS qui contient les séries composantes réconciliées, les totaux de contrôle réconciliés et toute variable spécifiée par l'énoncé ID. L'option OUT= n'est pas obligatoire. Si le nom de l'ensemble de données de sortie n'est pas spécifié, alors PROC TSRAKING le créera en utilisant les règles d'affectation des noms DATA_n.

TOLERANCEPERCENT=	TOLERANCEVALUE=
TOLP=nombre réel positif (incluant 0)	TOLV=nombre réel positif (incluant 0)

spécifie le niveau de tolérance, en pourcentage ou en valeur absolue, à utiliser pour le test ultime dans le cas où les totaux de contrôle sont contraignants (coefficient d'altérabilité = 0.0 pour les totaux de contrôle). Ce test compare les totaux de contrôle contraignants avec ceux calculés à partir des séries composantes réconciliées. Ce nombre est optionnel.

Par défaut :	TOLERANCEVALUE=1E-3.
Exemple :	Pour une tolérance de 1%, spécifiez TOLERANCEPERCENT =0.01 Pour une tolérance de 10, spécifiez TOLERANCEVALUE=10

TOLNEGRESULT=
TOLN=nombre réel négatif (excluant 0)

spécifie le seuil pour l'identification des valeurs négatives. Une valeur est considérée négative lorsqu'elle est inférieure à ce seuil. Ce nombre est optionnel. La valeur par défaut est -1E-3.

VERBOSE

affiche des résultats intermédiaires et le temps de CPU requis dans le journal (SAS log).

WARNNEGRESULT | NOWARNNEGRESULT

spécifie si la procédure doit afficher un message d'avertissement dans le journal (SAS log) lorsqu'une valeur négative créée par la procédure est inférieure au seuil tel que spécifié par l'option TOLNEGRESULT=. L'option par défaut est WARNNEGRESULT.

Énoncé ID

ID *variable-1 ... variable-n*;

Utilisez l'énoncé ID pour identifier des variables additionnelles à transférer du fichier d'entrée DATA= au fichier de sortie OUT=. Par défaut, le fichier de sortie ne contient que les variables identifiées dans les métadonnées (se référer à l'option METADATA=). Les variables spécifiées dans l'énoncé ID doivent être présentes dans le fichier d'entrée DATA=.

Détails

- À l'exception des variables spécifiées dans l'énoncé ID, une valeur manquante dans le fichier d'entrée DATA= entraînera l'arrêt de la procédure
 - Une valeur manquante ou négative dans le fichier d'entrée ALTERABILITY= entraînera l'arrêt de la procédure.
 - Des noms à deux niveaux tels que *libref.ensemble-de-données-SAS* peuvent être utilisés pour les fichiers d'entrée et pour rendre permanent le fichier de sortie.
-

Exemples

```
/* EXEMPLE 1 */
```

```
/* Dans cet exemple à 1 dimension, la somme des valeurs de "cars" et "vans" doit être  
égale à la valeur de "total" */
```

```
data myMetadata;  
input series $5.  
      Total1 $5.;  
datalines;  
cars total  
vans total  
;
```

```
data myData;  
input cars  
      vans  
      total;  
datalines;  
25 5 40  
;
```

```
proc tsraking  
  metadata=myMetadata  
  data=myData  
  out=outData;  
run;
```

```
/* EXEMPLE 2 */
```

```
/* Dans cet exemple à 2 dimensions, nous disposons d'estimés de ventes de "cars" et  
"vans" pour les régions "alb", "sask" and "man". La somme des 3 régions pour "cars" doit  
être égale à son total de contrôle (cars_alb + cars_sask + cars_man = cars_total). La  
somme des 3 régions pour "vans" doit aussi être égale à son total de contrôle (vans_alb +  
vans_sask + vans_man = vans_total). De plus, dans chaque region, la somme "cars" et
```

"vans" doit être égale au total de contrôle régional ($\text{cars_alb} + \text{vans_alb} = \text{alb_total}$; $\text{cars_sask} + \text{vans_sask} = \text{sask_total}$ and $\text{cars_man} + \text{vans_man} = \text{man_total}$). Finalement, les ventes de "vans" dans la région "sask" (vans_sask) ne sont pas altérables (coefficient d'altérabilité = 0) */

```
data myMetadata;
input series $9.
      Total1 $11.
      Total2 $11.;
datalines;
cars_alb cars_total alb_total
cars_sask cars_total sask_total
cars_man cars_total man_total
vans_alb vans_total alb_total
vans_sask vans_total sask_total
vans_man vans_total man_total
;

data myData;
input cars_alb cars_sask cars_man vans_alb vans_sask vans_man
      alb_total sask_total man_total cars_total vans_total;
datalines;
12 14 13 20 20 24 30 31 32 40 53
;

data myalter;
input cars_alb cars_sask cars_man vans_alb vans_sask vans_man
      alb_total sask_total man_total cars_total vans_total;
datalines;
1 1 1 1 0 1 0 0 0 0 0
;

proc tsraking
  metadata=myMetadata
  data=myData
  alterability=myAlter
  out=outData;
run;
```

/* EXEMPLE 3 */

/* Dans cet exemple à 2 dimensions avec contraintes temporelles, nous avons les contraintes transversales $A_1 + A_2 + A_3 = A$; $B_1 + B_2 + B_3 = B$; $A_1 + B_1 = _1$; $A_2 + B_2 = _2$ et $A_3 + B_3 = _3$. Nous avons 4 observations (trimestres) pour chaque valeur et nous voulons préserver les totaux annuels. Pour chaque variable, la somme des 4 valeurs trimestrielles doit demeurer la même après l'étape de réconciliation. */

```
data mymetadata;
input series $3.
      total1 $3.
      total2 $3.;
datalines;
A_1 _1 A
B_1 _1 B
A_2 _2 A
B_2 _2 B
A_3 _3 A
B_3 _3 B
;

data mydata;
input A_1 A_2 A_3 B_1 B_2 B_3 A B _1 _2 _3;
```

```

datalines;
12 14 13 20 20 24 40 53 30 31 32
10 9 15 21 29 20 25 80 35 35 35
12 8 17 15 20 30 40 59 23 32 44
9 9 14 17 24 23 37 71 28 35 45
;

proc tsraking
  metadata=myMetadata
  data=myData
  out=outData;
run;

```

Notes

Ce document est un guide pour l'utilisation de la procédure TSRAKING. PROC TSRAKING fait partie du logiciel G-Series de Statistique Canada, précédemment appelé Forillon.

Pour plus de renseignements, veuillez vous adresser à l'équipe de soutien de G-Series à l'aide de l'adresse courriel de [G-Series](mailto:G-Series@statcan.gc.ca) (G-Series@statcan.gc.ca) ou consultez le site web de [G-Series](#) (Intranet de Statistique Canada seulement).

Bibliographie

Bérubé, J. et S. Fortier (2009). « PROC TSRAKING: An in-house SAS procedure for balancing time series », recueil de la conférence **JSM 2009**, Business and Economic Section. Alexandria, VA: American Statistical Association.

Dagum, E. B. et P. Cholette (2006). **Benchmarking, Temporal Distribution and Reconciliation Methods of Time Series**. Springer-Verlag, New York, Lecture Notes in Statistics, Vol. 186

Fortier, S. et B. Quenneville (2009). « Reconciliation and Balancing of Accounts and Time Series », recueil de la conférence **JSM 2009**, Business and Economic Section. Alexandria, VA: American Statistical Association.
